

Northeastern University
College of Computer and Information Science

CS1100: Computer Science and Its Applications

Excel Basics

Leena Razzaq
lrazzaq@ccs.neu.edu


Modified from originals created by Martin Schedlbauer, Peter Douglass and Peter Golbus

Spreadsheets

- Spreadsheets are among the most useful technical business applications.
- Principally used for calculations and manipulation of tabular data.
- Common spreadsheet applications:
 - Microsoft Excel
 - Google Spreadsheet
 - Zoho Sheet


Spreadsheet Layout

- Tabular layout arranged in rows and columns.
 - Columns are labeled with letters
 - Rows are labeled with numbers
- Cells are at the intersection of rows and columns
 - Example cell reference: A3, C9
- Cells can contain:
 - Numbers, dates, text, or other data
 - Formulas using functions and cell references



	A	B	C
1			
2			
3			
4			
5			
6			
7			
8			
9			
10			
11			
12			
13			
14			
15			
16			
17			
18			
19			
20			
21			
22			
23			
24			

Microsoft Excel 2010



Cell Ranges

- Many functions require cell ranges:
 - Column Range: A1:A10
 - Row Range: A5:K5
 - Matrix: A1:C5

Functions

- Excel provide thousands of functions to build spreadsheet models:
 - Financial, *e.g.*, pmt, irr, fv, db
 - Aggregation, *e.g.*, sum, count, average
 - Text, *e.g.*, left, mid, trim
 - Date & Time, *e.g.*, today, time, second
 - Lookup, *e.g.*, choose, vlookup, match
 - Logical, *e.g.*, if, not, or
 - Statistical, *e.g.*, median, correl
 - Engineering, *e.g.*, bessell, imlog
 - Trigonometric, *e.g.*, sin, tan, acos

Entering Formulas and Functions

- To enter formulas and functions:
 - start entry with =
- Example:


	A	B
1	Expenses	2010
2	Rent	500
3	Equipment	100
4	Wages	400
5	Total	=sum(b2:b4)

Cell References in Functions

- Most functions require parameters.
- To keep your model flexible and correct even when the data changes, only use cell references in functions.

	A	B
1	Interest Rate:	0.0475
2	Principal:	10000
3	Term:	72
4	Monthly Payment:	=PMT(B1/12,B3,B2)

Copying Cells

- To copy cells:
 - CTRL+C to copy and CTRL+V to paste
 - or
 - Use cell dragging 
- Cell references are automatically adjusted when copied.
- Cell references can be locked \$.
 - \$A1:\$A5 is not adjusted when column copied
 - A\$1:C\$1 is not adjusted when row copied
 - \$A\$1 is never adjusted when copied

Anchors and Cell Dragging

- Formulas can be copied to adjacent cells by dragging.
- Dragging changes cell references.
- Usually this is what you want, but sometimes it breaks your formulas.
- Anchors (\$) stop cell references from changing.
- But when do you use them?

When to Anchor

- Manually rewrite your formula in the cell to the right and the cell below the original cell.
- Compare the *letters* in the original formula to the letters in the formula to the *right*.
- If a letter didn't change, put a \$ before it in the original cell.

When to Anchor

- Manually rewrite your formula in the cell to the right and the cell below the original cell.
- Compare the *numbers* in the original formula to the letters in the formula to *below*.
- If a number didn't change, put a \$ before it in the original cell.

Demo: Copying Formulas

	A	B	C
1	Expenses	2010	2011
2	Rent	500	500
3	Equipment	100	100
4	Wages	400	400
5	Total	=SUM(B2:B4)	


	A	B	C
1	Expenses	2010	2011
2	Rent	500	500
3	Equipment	100	100
4	Wages	400	400
5	Total	=SUM(B2:B4)	=SUM(C2:C4)

- Notice what happens to the cell references when copying from row to row or column to column.
- Observe the effect of cell locking with \$.

[Click to watch video demonstration](#)

Named Ranges

- To make your formulas easier to read, use named cell ranges.
- To create a named range:
 - Highlight cells to include in named range
 - Click right mouse button on any cell in the selected range for context menu
 - Choose “Define Name...” and provide name
- **Note: named ranges are never adjusted when row or column copied, i.e. both cells and columns are automatically anchored in named ranges.**



[Click here to watch demonstration...](#)

Named Ranges in Functions


- Named ranges can make function parameters easier to understand:

	A	B
1	Interest Rate:	0.0475
2	Principal:	10000
3	Term:	72
4	Monthly Payment:	=PMT(InterestRate/12,TermInMonths,Principal)

Managing Named Ranges

- To manage (delete, edit, rename, etc) named ranges
 - In the Formulas ribbon
 - Click on Name Manager

Managing Named Ranges



Showing Formulas

- To show the formulas in your spreadsheet, press CTRL+~.

	A	B
1	Expenses	
2	Rent	500
3	Equipment	100
4	Wages	400
5	Total	<u>1000</u>

	A	B
1	Expenses	
2	Rent	500
3	Equipment	100
4	Wages	400
5	Total	<u>=SUM(B2:B4)</u>

Formatting

- Formatting changes the way values are displayed, but does not change the actual value being used in functions.
- Common formatting options:
 - Currency values
 - Time and date values
 - Numeric formats and decimal points
 - Percent

Demo: Formatting

	A	B	C
1	Interest Rate:	0.0475	
2	Principal:	10000	
3	Term:	72 months	
4	Monthly Payment:	-159.8922446	

Unformatted values

	A	B	C
1	Interest Rate:	4.75%	<i>Percent</i>
2	Principal:	\$ 10,000.00	<i>Currency</i>
3	Term:	72 months	
4	Monthly Payment:	\$ (159.89)	<i>Accounting</i>



Formatted values

[Click here to watch demonstration...](#)


Conditional Formatting

- Conditional formatting is allows the application of specified formatting only when certain conditions are met.
- On the Home tab, in the Styles group, click the arrow next to Conditional Formatting

Stock	Opening	Closing	Change
DNDN	5.88	6.02	0.14
FB	27.77	28.76	0.99
MSFT	27.25	26.74	-0.51
BAC	11.96	12.11	0.15




Formatting Example



Look Carefully at the Formatting Example

- It *appears* to say that the sum of
3.05
2.02
1.03
is 6.11
- That is not mathematically correct!
- The next slide shows what is actually in each cell. (Control / ~)


Formatting Example with Control / ~




What happened?

- What happened is that the cells in column 'A' were formatted to show only 2 digits after the decimal point.
- However, the numbers in column 'A' actually had more than 2 digits after the decimal point. The last digits were “hidden”.
- Hiding some of the digits can yield results that appear to be wrong.

Rounding Example



Rounding Example with Control / ~



Rounding Example

- The addition in column 'B' is now mathematically correct.
- The value that appears in cell B5 is in fact the sum of the numbers appearing in cells B1:B3

Warning!

- The values ***displayed*** by a computer application are often not the values that are ***used*** inside that application.
- If you need a value that has only a certain number of digits after the decimal place, you must ***round*** that value, not merely ***format*** it.
- Excel: ROUND(Range, Decimal Places)

Bottom Line

- If a computer's arithmetic appears incorrect, it is may be a formatting / rounding error.
- Excel: fix these errors with the ROUND function.

Hiding Columns or Rows

- To make spreadsheets easier to read, you may wish to hide rows or columns that contain auxiliary (or supporting) values or temporary calculations.
- Right-Click on the row or column header and select “Hide”.

[Click here to watch demonstration...](#)

The IF Function

- The IF function allows a cell to be filled with one of two possible values.

- General form of IF:

`=IF(condition,value_if_true,value_if_false)`

- Example:

	A	B
1	<i>Tax Exempt?</i>	Yes
2	<i>Sales Tax:</i>	5.65%
3	<i>Order Total:</i>	\$100.00
4	<i>Tax Due:</i>	=IF(B1="Yes",0,B2*B3)

Cell B4 is either \$0 if the customer is tax exempt or the tax due is the order total multiplied by the tax rate.

Taking a Closer Look at IF

	A	B
1	<i>Tax Exempt?</i>	Yes
2	<i>Sales Tax:</i>	5.65%
3	<i>Order Total:</i>	\$100.00
4	<i>Tax Due:</i>	


Does cell *B1*
contain "Yes"?

If so, then cell *B4* will be
filled with the value 0

=IF(B1="Yes", 0, B2*B3)

If not, then cell *B4* will
be filled with the result
of the formula *B2*B3*

A Closer Look at the Statement



How Does **IF** Work?

- **IF** does not perform any calculation, it simply fills a cell with one of two values.
- The values can be:
 - literals (actual numbers or text), *e.g.*, 0
 - results of functions or formulas
 - empty cells (""), are two double-quotes
 - some cell reference, *e.g.*, B2

The **IF** Condition

- The **IF** condition is a logical expression, I.E. it evaluates to true or false.
- Examples:
 - equality (=)
 - less than (<) or less than or equal (<=)
 - greater than (>) or greater than or equal (>=)
- Complex conditions can be built with the **AND** and **OR** functions.

A Complete Spreadsheet Model

- Spreadsheet to calculate the market value of a precious metals portfolio.

	A	B	C	D	E
1	Market Price of Silver per OZ:			\$ 30.00	
2	Market Price of Gold per OZ:			\$1,401.00	
3					
4	Metal	Amt (oz)	Date	Basis (oz)	Gain/Loss
5	Gold	10.00	5/7/2008	\$ 392	\$ 10,090
6	Silver	100	9/1/2009	13.75	\$ 1,625
7	Gold	2	9/5/2009	899	\$ 1,004
8					
9	Total Portfolio Value:			\$ 12,719	
10	Total Gold Value:			\$11,094.00	
11	Total Silver Value:			\$ 1,625.00	

=B5*(IF(A5="Gold",\$D\$2,\$D\$1)-D5)

=SUM(E5:E7)

=SUMIFS(E5:E7,A5:A7,"Gold")

Filtering Data

- An alternative way of summing data.

	A	B	C	D	E	F	G
1	Market Price of Silver per OZ:			\$ 30.00			
2	Market Price of Gold per OZ:			\$1,401.00			
3							
4	Metal	Amt (oz)	Date	Basis (oz)	Gain/Loss	Gold	Silver
5	Gold	10.00	5/7/2008	\$ 392	\$ 10,090	\$ 10,090	\$ -
6	Silver	100	9/1/2009	13.75	\$ 1,625	\$ -	\$ 1,625
7	Gold	2	9/5/2009	899	\$ 1,004	\$ 1,004	\$ -
8							
9	Total Portfolio Value:				\$ 12,719	\$ 11,094	\$ 1,625

=IF(\$A5=F\$4,\$E5,0)

=SUM(F5:F7)

 [Try it out online with Zoho Sheet](#)

COUNTIF

- COUNTIF allows you to display the number of cells in a range whose values meets specific criteria.
- The syntax of the COUNTIF function is:
COUNTIF(**range,criteria**) ... where range is a group of cells, and criteria is the value a cell must have to be counted.
- The default operator for criteria is "equals" and should not be specified.
- Operators (>, <, >=, <=, <> and =) must be enclosed in quotation marks and <> means "not equal".

COUNTIF

- Example:

D6		fx		=COUNTIF(D2:D5, ">.2")		
	A	B	C	D	E	F
1	Stock	Opening	Closing	Change		
2	DNDN	5.88	6.02	0.14		
3	FB	27.77	28.76	0.99		
4	MSFT	27.25	26.74	-0.51		
5	BAC	11.96	12.11	0.15		
6				1		

Excel Basics

FLEXIBLE MODELS

Assume Data May Change!

- It is best practice to write your spreadsheets in such a way that they give correct results for given data, regardless of what that data might be, and not merely the correct results for a particular data set.
- If the data changes, the answer should be correct for the new data set.

Don't Duplicate Data!

- Since the given data for a problem may change, this data should appear as given data in one place only.
- If data given for a problem is repeated in many places, then changing that data will require changes in many places.
 - This can be a source of errors as well as a source of unnecessary work.

Building Flexible Models

- Don't use actual numbers or text in your formulas and functions, i.e., don't “hard code” values.
- To keep your model general and flexible when data change, use only cell references.

Inductively Defined Problems

- An inductively defined problem has 2 parts:
 - A set of starting conditions
 - A set of rules that describe how data changes from one step to the next.

Inductively Defined Problems


- Example:

This is the starting condition



An arrow points from the box containing 'This is the starting condition' to the first bullet point, '– You deposit \$1000 in a savings account'.

- You deposit \$1000 in a savings account
- At the end of each year, you receive 4% interest on the balance in your account.



An arrow points from the box containing 'This rule describes how data changes from one step to the next' to the second bullet point, '– At the end of each year, you receive 4% interest on the balance in your account.'

This rule describes how data changes from one step to the next

Book1 - Microsoft Excel

Home Insert Page Layout Formulas Data Review View Acrobat

Clipboard Font Alignment Number Styles Cells Editing

R11 fx

	A	B	C	D	E	F	G	H	I	J	K	L
1	Initial Deposit	\$1,000.00		Year	Ending Balance	Interest						
2	Interest rate	4.00%		0	\$1,000.00							
3				1	\$1,040.00	\$40.00						
4				2	\$1,081.60	\$41.60						
5				3	\$1,124.86	\$43.26						
6				4	\$1,169.86	\$44.99						
7				5	\$1,216.65	\$46.79						
8				6	\$1,265.32	\$48.67						
9				7	\$1,315.93	\$50.61						
10												
11												
12												

Sheet1 Sheet2 Sheet3

Ready 100%

Book1 - Microsoft Excel

Home Insert Page Layout Formulas Data Review View Acrobat

Clipboard Font Alignment Number Styles Cells Editing

R11

	A	B	C	D	E	F
1	Initial Deposit	1000		Year	Ending Balance	Interest
2	Interest rate	0.04		0	=B1	
3				=D2+1	=E2+F3	=E2*\$B\$2
4				=D3+1	=E3+F4	=E3*\$B\$2
5				=D4+1	=E4+F5	=E4*\$B\$2
6				=D5+1	=E5+F6	=E5*\$B\$2
7				=D6+1	=E6+F7	=E6*\$B\$2
8				=D7+1	=E7+F8	=E7*\$B\$2
9				=D8+1	=E8+F9	=E8*\$B\$2
10						
11						
12						

Sheet1 Sheet2 Sheet3

Ready 100%

Inductively Defined Problems

- The parameters of our problem appear separately from the calculations that model our problem (columns A and B).
- There is a special year 0 in our solution that serves as a place holder for our starting condition.
- The formulas for the data for each year other than year 0 are similar, differing only in the cells that they reference.